



**UNIVERSIDAD
DE GRANADA**

Prácticas de Scilab
Reactor Discontinuo Mezcla Perfecta (RDMP)

Christian Pérez Hita

Pregunta 1: RDMP-1

En un reactor discontinuo mezcla perfecta isoterma, se lleva a cabo la siguiente reacción química: $A \rightarrow B$
 Constante de velocidad = 0.5 h^{-1} El reactor se pone en marcha con 1 mol/L de A.

- (A) Obtener la dinámica durante 5 h de reacción, es decir, la evolución temporal de las concentraciones de A y B, así como de la conversión

Solución:



```

1  clear; clc;
2  // RDMP-1.sce
3  // A => B
4
5  // Isoterma
6
7  //! acumulacion v*ca
8
9  /* SISTEMA DE ECUACIONES DIFERENCIALES
10 function dxdt = f(t,x)
11     // Variables diferenciales
12     CA = x(1)
13     CB = x(2)
14     // Velocidad de reacción
15     r = k*CA
16     // Balance de materia para A
17     // d(V*CA)dt = -r*V
18     dCAdt = -r
19     // Balance de materia para B
20     // d(V*CB)dt = r*V
21     dCBdt = r
22     // Derivadas
23     dxdt(1) = dCAdt
24     dxdt(2) = dCBdt
25 endfunction
26
27 // CONSTANTES
28 k = 0.5; // h-1
29
30 // CONDICIONES INICIALES
31 CAini = 1; CBini = 0; // mol/L
32 xini = [CAini;CBini];
33
34 // TIEMPO
35 tfin = 5; dt = 0.01; t = 0:dt:tfin; // h-1
36
37 // RESOLVER
38 x = ode(xini,0,t,f);
39 CA = x(1,:); CAfin = CA($)
40 CB = x(2,:); CBfin = CB($)
41
42 XA = 1 - CA/CAini; XAfin = XA($)
43
44 // GRÁFICAS
45 scf(1); clf(1);
46 plot(t,CA,t,CB);
47 xgrid; xlabel('t'); legend('CA','CB',-2,%f);
48
49 scf(2); clf(2);
50 plot(t,XA,'m-');

```

```

51 xgrid; xlabel('t'); legend('XA',-2,%f);
52
53 disp('Concentracion final de A: '+string(CAfin)+' mol/L');
54 disp('Concentracion final de B: '+string(CBfin)+' mol/L');
55 disp('Conversión final de A: '+string(XAfin)+' mol/L');

```

Nota:-

En resultado se muestra lo que aparece en pantalla al ejecutar el código:

Resultado:

Concentración final de A: 0.082085 mol/L
 Concentración final de B: 0.917915 mol/L
 Conversión final de A: 0.917915 mol/L




Figura 1: Evolución temporal del reactor por lotes

Pregunta 2: RDMP-2

La reacción química $A \rightarrow B$

Factor preexponencial = $4.15E5 \text{ s}^{-1}$

Energía de activación = 11200 cal/mol

Entalpía de reacción = -50400 cal/mol

Se lleva a cabo en un reactor discontinuo mezcla perfecta adiabático que se pone en marcha con 0.5 mol/L de A a 285 K. Considerando para la mezcla reaccionante una capacidad calorífica de 0.9 cal/(g*K) y una densidad de 1070 g/L:

- (A) Obtener la dinámica del proceso
- (B) Determinar la temperatura cuando se alcanza el 90 % de conversión

Solución:

```

1 // d(V*CA)dt = -r*V consume a por eso es -r y el reacto de a es r lo consumido *v
2 // v es constante por lo que puede salir de la derivada
3 // dCAdt = -r
4 // Variables diferenciales como ya tenemos CA en una ec definida la ponemos como constante
5 // d(V*RHO*CP*T)dt //<... (m*cp*t)//= -H*r*V ; el signo menos de la entalpia es por que al ser
6 // exotermica por convenio la h es negativa y como tiene que aumentar
7 // la temperatura es necesario para --=* // dTdt = -H*r/(RHO*CP)
8
9 // si la reaccion es irreversible no se suele poner los balances de materia del producto
10 // se puede hacer aunque no sea necesario ya que la potencia de calculo no la complica
11
12 // en los ejercicios nunca vamos a tener que realizar cambios de unidades pero si poner las unidades
13 // constante que buscar siempre la r es decir tenerlas tabuladas
14
15 // el tiempo no nos lo da el ejercicio ponemos un numero provisional de prueba y error
16
17 // xgrid; xlabel('t'); legend('CA',-2,%f);
18 // xgrid; xlabel('t'); legend('XA',-2,%f);
19 // plot(t,T,'r-',tXAobj,txAobj,'ro');
20 // xgrid; xlabel('t'); legend('T',-2,%f);// el ,-2 es la posicion arriba izquierda fuera
21 // el -3 es abajo izquierda fuera del grafico , el %f es para eliminar la caja de la leyenda
22
23
24 clear; clc;
25 // RDMP-2.sce
26 // A => B
27 // Adiabático
28
29 // SISTEMA DE ECUACIONES DIFERENCIALES
30 function dxdt = f(t, x)
31     CA = x(1);
32     T = x(2);
33
34     // Ecuación de Arrhenius
35     k = k0 * exp(-E / (R * T));
36
37     // Velocidad de reacción
38     r = k * CA;
39
40     // Balance de materia para A
41     dCAdt = -r;
42
43     // Balance de energía
44     dTdt = -H * r / (RHO * CP);
45
46     // Retornar como columna
47     dxdt = [dCAdt; dTdt];
48 endfunction
49
50 // CONSTANTES
51 CP = 0.9; // cal/(g*K)
52 RHO = 1070; // g/L
53 H = -50400; // cal/mol /* Reacción exotérmica ya que la entalpia de reaccion tiene signo negativo
54 k0 = 4.15E5; // s-1
55 E = 11200; // cal/mol
56 R = 1.987; // cal/(mol*K)
57
58
59

```

```

60 // CONDICIONES INICIALES
61 CAini = 0.5; // mol/L
62 Tini = 285; // K
63 xini = [CAini; Tini];
64
65 // TIEMPO INICIAL
66 tfin = 50; dt = 1;
67 XAobj = 0.90; // Conversión objetivo
68 XAalcanzado = 0; // Variable para comprobar si se alcanza XAobj
69 tfinal_max = 10000; // Tiempo máximo permitido
70 t = 0:dt:tfin;
71
72 // SOLVER
73 while (XAalcanzado < XAobj & tfin < tfinal_max)
74     x = ode(xini, 0, t, f);
75
76     // Extraer valores de concentración y temperatura
77     CA = x(1, :);
78     T = x(2, :);
79
80     // Calcular conversión
81     XA = 1 - CA / CAini;
82
83     // Revisar si se alcanzó el valor objetivo
84     if max(XA) >= XAobj then
85         XAalcanzado = max(XA);
86         break; // Termina el bucle si se alcanza la conversión objetivo
87     else
88         // Aumentar el tiempo de simulación si no se alcanza XAobj
89         tfin = tfin + 50;
90         t = 0:dt:tfin;
91     end
92 end
93
94 // Encontrar el tiempo en que se alcanza XAobj
95 indexXAobj = find(XA >= XAobj, 1);
96 if indexXAobj <> [] then
97     tXAobj = t(indexXAobj);
98     TXAobj = T(indexXAobj);
99 else
100     tXAobj = NaN;
101     TXAobj = NaN;
102 end
103
104 // GRÁFICAS
105 scf(1); clf(1);
106 plot(t, CA, 'b');
107 xgrid; xlabel('t'); legend('CA', -2, %f);
108
109 scf(2); clf(2);
110 plot(t, XA, 'm-', tXAobj, XAobj, 'mo');
111 xgrid; xlabel('t'); legend('XA', -2, %f);
112
113 scf(3); clf(3);
114 plot(t, T, 'r-', tXAobj, TXAobj, 'ro');
115 xgrid; xlabel('t'); legend('T', -2, %f);
116 disp('la conversión de A es de '+string(XAalcanzado*100)+'%');
117 disp('la temperatura en la que se alcanza el 90% de conversión es de '+string(TXAobj)+' K');
118 disp('el tiempo en el que se alcanza el 90% de conversión es de '+string(tXAobj)+' s');
119 disp('el tiempo máximo permitido es de '+string(tfinal_max)+' s');

```

Resultado:

la conversión de A es de 90.551024 %
 la temperatura en la que se alcanza el 90 % de conversión es de 308.55992 K
 el tiempo en el que se alcanza el 90 % de conversión es de 839 s
 el tiempo máximo permitido es de 10000 s






Figura 2: Concentración de A



(a) Conversión de A



(b) Temperatura

Figura 3: Evolución de la conversión y temperatura en el reactor

Pregunta 3: RDMP-3a

La reacción química $A \rightarrow B$

Factor preexponencial = $2.2E4 \text{ s}^{-1}$

Energía de activación = 41570 J/mol

Entalpía de reacción = $-5E5 \text{ J/mol}$

Se lleva a cabo durante 1500 s en un reactor discontinuo mezcla perfecta de 1 m^3 que dispone de una camisa de 4 m^2 a 283 K con un coeficiente global de transmisión de calor de $400 \text{ J/(m}^2 \cdot \text{s} \cdot \text{K)}$. Considerando para la mezcla reaccionante una densidad de 980 kg/m^3 y una capacidad calorífica de $4200 \text{ J/(kg} \cdot \text{K)}$:

- (A) Obtener la dinámica del proceso si el reactor se pone en marcha con 500 mol/m^3 de A a 283 K
- (B) Localizar el instante de temperatura máxima

Solución:



```

1
2  clear; clc;
3  // RDMP-3a.sce
4  // A => B
5  // No adiabático: camisa a temperatura constante
6
7  // SISTEMA DE ECUACIONES DIFERENCIALES
8  function dxdt = f(t,x)
9      // Variables diferenciales
10     CA = x(1)
11     T  = x(2)
12     // Ecuación de Arrhenius
13     k = k0*exp(-E/(R*T))
14     // Velocidad de reacción
15     r = k*CA
16     // Calor transferido del reactor a la camisa
17     Q = U*A*(T-TJ)
18     // Balance de materia para A
19     // d(V*CA)dt = -r*V
20     dCAdt = -r
21     // Balance de energía no hay entrada ni salida solo generacion
22     //consideracion camisas en reactores y serpentines refrigeran por ello el signo -
23     // y ponemos -Q que hay que calcularlo (línea 16)
24     // d(V*RHO*CP*T)dt = -H*r*V - Q
25     dTdt = -H*r/(RHO*CP) - Q/(V*RHO*CP)
26     // Derivadas
27     dxdt(1) = dCAdt
28     dxdt(2) = dTdt
29 endfunction
30
31 // CONSTANTES
32 V = 1; // m3
33 RHO = 980; // kg/m3
34 CP = 4200; // J/(kg*K)
35 U = 400; // J/(m2*s*K)
36 A = 4; // m2
37 TJ = 283; // K
38 H = -5E5; // J/mol
39 k0 = 2.2E4; // s-1
40 E = 41570; // J/mol
41 R = 8.314; // J/(mol*K)
42
43 // CONDICIONES INICIALES
44 CAini = 500; // mol/m3

```

```


45 Tini = 283; // K
46 xini = [CAini; Tini];
47
48 // TIEMPO
49 tfin = 1500; dt = 1; t = 0:dt:tfin; // s
50
51 // RESOLVER
52 x = ode(xini,0,t,f);
53 CA = x(1,:); CAfin = CA($)
54 T = x(2,:); Tfin = T($)
55
56 [Tmax,indexTmax] = max(T)
57 tTmax = t(indexTmax)
58 disp(' La temperatura maxima es: '+string(Tmax)+' K')
59 disp(' El tiempo en el que se alcanza la temperatura maxima es: '+string(tTmax)+' s')
60
61 // GRÁFICAS
62 scf(1); clf(1);
63 plot(t,CA);
64 xgrid; xlabel('t'); legend('CA',-2,%f);
65
66 scf(2); clf(2);
67 plot(t,T,'r-',tTmax,Tmax,'ro');
68 xgrid; xlabel('t'); legend('T',-2,%f);

```


Resultado:

La temperatura maxima es: 329.26838 K

El tiempo en el que se alcanza la temperatura maxima es: 1191 s



(a) Conversión de A



(b) Temperatura

Figura 4: Evolución de la conversión y temperatura en el reactor

Pregunta 4: RDMP-3b

La reacción química $A \rightarrow B$

Factor preexponencial = $2.2E4 \text{ s}^{-1}$

Energía de activación = 41570 J/mol

Entalpía de reacción = $-5E5 \text{ J/mol}$

se lleva a cabo durante 1500 s en un reactor discontinuo mezcla perfecta de 1 m^3 que dispone de una camisa de 0.1 m^3 y 4 m^2 capaz de transmitir calor con un coeficiente global de $400 \text{ J/(m}^2 \cdot \text{s} \cdot \text{K)}$ y por la que circulan $0.001 \text{ m}^3/\text{s}$ de agua que entran a 283 K . Considerando para la mezcla reaccionante una densidad de 980 kg/m^3 y una capacidad calorífica de $4200 \text{ J/(kg} \cdot \text{K)}$:

- (A) Obtener la dinámica del proceso si el reactor se pone en marcha con 500 mol/m^3 de A a 283 K
- (B) Determinar el caudal de agua que debe circular por la camisa para que la temperatura del reactor no supere en ningún momento los 330 K

```

1  clear; clc;
2  // RDMP-3b.sce
3  // A => B
4  // No adiabático: camisa a temperatura variable
5
6  // SISTEMA DE ECUACIONES DIFERENCIALES
7  function dxdt = f(t,x)
8      // Variables diferenciales
9      CA = x(1)
10     T = x(2)
11     TJ = x(3)
12     // Ecuación de Arrhenius
13     k = k0*exp(-E/(R*T))
14     // Velocidad de reacción
15     r = k*CA
16     // Calor transferido del reactor a la camisa
17     Q = U*A*(T-TJ)
18     // Balance de materia para A (no hace falta el balance para b)
19     // d(V*CA)dt = -r*V
20     dCAdt = -r
21     // Balance de energía en el reactor
22     // d(V*RHO*CP*T)dt = -H*r*V - Q
23     dTdt = -H*r/(RHO*CP) - Q/(V*RHO*CP)
24     // Balance de energía en la camisa
25     // d(VJ*RHOJ*CPJ*TJ)dt = FJ*RHOJ*CPJ*TJ0 - FJ*RHOJ*CPJ*TJ + Q
26     //hay 2 terminos sale la constante por lo que queda la derivada de TJ que es x(3)
27     dTJdt = FJ*(TJ0-TJ)/VJ + Q/(VJ*RHOJ*CPJ)
28     // Derivadas
29     dxdt(1) = dCAdt
30     dxdt(2) = dTdt
31     dxdt(3) = dTJdt
32 endfunction
33
34 // CONSTANTES
35 V = 1; // m3
36 RHO = 980; // kg/m3
37 CP = 4200; // J/(kg*K)
38 U = 400; // J/(m2*s*K)
39 A = 4; // m2
40 VJ = 0.1; // m3
41 FJ = 6.15E-3; //m3/s
42 TJ0 = 283 // K
43 RHOJ = 1000; // kg/m3
44 CPJ = 4180; //J/(kg*K)
45 H = -5E5; // J/mol

```


```

46 k0 = 2.2E4; // s-1
47 E = 41570; // J/mol
48 R = 8.314; // J/(mol*K)
49
50 // CONDICIONES INICIALES
51 CAini = 500; // mol/m3
52 Tini = 283; // K
53 TJini = 283; // K
54 //el valor inicial de la camisa no nos lo dice asi que supones 283
55 //asumimos que inicialmente el reactor y la camisa estan a la misma T
56 xini = [CAini; Tini; TJini];
57
58 // TIEMPO
59 tfin = 1500; dt = 1; t = 0:dt:tfin; // s
60
61 // RESOLVER
62 x = ode(xini,0,t,f);
63 CA = x(1,:); CAfin = CA($);
64 T = x(2,:); Tfin = T($);
65 TJ = x(3,:); TJfin = TJ($);
66 disp("La concentracion final de A es: "+ string(CAfin) + " mol/m3")
67 disp("La temperatura final del reactor es: "+ string(Tfin) + " K")
68 disp("La temperatura final de la camisa es: "+ string(TJfin) + " K")
69 [Tmax,indexTmax] = max(T);
70 tTmax = t(indexTmax);
71 disp("La temperatura maxima es: "+ string(Tmax) + " K" + " y se alcanza en el instante " + string(tTmax) + " s")
72 // GRÁFICAS
73 scf(1); clf(1);
74 plot(t,CA);
75 xgrid; xlabel('t'); legend('CA',-2,%f);
76
77 scf(2); clf(2);
78 plot(t,T,'r-',t,TJ,'r:',tTmax,Tmax,'ro');
79 xgrid; xlabel('t'); legend('T','TJ',-2,%f);
80
81 scf(3);
82 plot(FJ,Tmax,'ro');
83 xgrid; xlabel('FJ'); ylabel('Tmax');
84
85 //para saber el caudal de tj lo cambiamos pot tanteo por intuicion ya que queremos que tmax sea la
86 //t final o haciendo un bucle probar en casa bucle para calcular el caudal necesario
87 //que mantenga el reactor por debajo de 330°
88 //if tmax> T then Fj=FJi+0.01meter tambien un epsilon <E2 como criterio de parada /break
89 //y un numero maximo de iteraciones n entre la linea 33 y 69


```

Resultado:

La concentracion final de A es: 4.4895026 mol/m³
 La temperatura final del reactor es: 327.25683 K
 La temperatura final de la camisa es: 285.605 K
 La temperatura maxima es: 329.99939 K y se alcanza en el instante 1192 s



(a) Conversión de A vs t



(b) Temperaturas vs t




Figura 6: Temperatura

Pregunta 5: RDMP-3c

La reacción química $A \rightarrow B$

Factor preexponencial = $2.2E4 \text{ s}^{-1}$

Energía de activación = 41570 J/mol

Entalpía de reacción = $-5E5 \text{ J/mol}$

se lleva a cabo durante 1500 s en un reactor discontinuo mezcla perfecta de 1 m^3 que dispone de un serpentín de 13 m de longitud y 0.1 m de diámetro capaz de transmitir calor con un coeficiente global de $400 \text{ J/(m}^2 \cdot \text{s} \cdot \text{K)}$ y por el que circulan $0.001 \text{ m}^3/\text{s}$ de agua que entran a 283 K. Considerando para la mezcla reaccionante una densidad de 980 kg/m^3 y una capacidad calorífica de $4200 \text{ J/(kg} \cdot \text{K)}$:

- Ⓐ Obtener la dinámica del proceso si el reactor se pone en marcha con 500 mol/m^3 de A a 283 K

```
1 clear; clc;
2 // RDMP-3c.sce
3 // A => B
```

```

4  // No adiabático: serpentín
5  //matiz ejercicio difícil
6  // SECTORES
7  N = 10; // divisiones del serpentín
8
9  // SISTEMA DE ECUACIONES DIFERENCIALES
10 function dxdt = f(t,x)
11     // Variables diferenciales
12     Ts = x(1:N)
13     CA = x(N+1)
14     T = x(N+2)
15     // Calor transferido del reactor al serpentín
16     i = 1:N; Q(i) = U*As*(T-Ts(i));
17     // Balance de energía en el serpentín
18     //  $d(Vs \cdot RHOs \cdot CPs \cdot Ts(i))dt = Fs \cdot RHOs \cdot CPs \cdot Ts(i-1) - Fs \cdot RHOs \cdot CPs \cdot Ts(i) + Q(i)$ 
19     // Sector 1 como no puede ser i-1 tiene su propio balance abajo
20     dTsdt(1) = Fs*(Ts0-Ts(1))/Vs + Q(1)/(Vs*RHOs*CPs)
21     // Resto de sectores
22     i = 2:N; dTsdt(i) = Fs*(Ts(i-1)-Ts(i))/Vs + Q(i)/(Vs*RHOs*CPs)
23     // Ecuación de Arrhenius
24     k = k0*exp(-E/(R*T))
25     // Velocidad de reacción
26     r = k*CA
27     // Balance de materia para A
28     //  $d(V \cdot CA)dt = -r \cdot V$ 
29     dCAdt = -r
30     // Balance de energía en el reactor
31     //  $d(V \cdot RHO \cdot CP \cdot T)dt = -H \cdot r \cdot V - Q$ 
32     dTdt = -H*r/(RHO*CP) - sum(Q)/(V*RHO*CP)
33     // Derivadas
34     dxdt(1:N) = dTsdt
35     dxdt(N+1) = dCAdt
36     dxdt(N+2) = dTdt
37 endfunction
38
39 // CONSTANTES
40 V = 1; // m3
41 RHO = 980; // kg/m3
42 CP = 4200; // J/(kg*K)
43
44 U = 400; // J/(m2*s*K)
45 Ds = 0.1; // m
46 Lstot = 13; // m
47 Vstot = %pi/4*Ds^2*Lstot // m3 (volumen de sector total)
48 Astot = %pi*Ds*Lstot // m2
49 Vs = Vstot/N; // m3 (volumen de sector)
50 As = Astot/N; // m2 (area de sector)
51 Fs = 1E-3; // m3/s (flujo serpentín)
52 Ts0 = 283; // K
53 RHOs = 1000; // kg/m3
54 CPs = 4180; // J/(kg*K)
55
56 H = -5E5; // J/mol
57 k0 = 2.2E4; // s-1
58 E = 41570; // J/mol
59 R = 8.314; // J/(mol*K)
60
61 // CONDICIONES INICIALES
62 i = 1:N; Tsini(i) = 283; // K iniciando todos los sectores a t inicial
63 CAini = 500; // mol/m3
64 Tini = 283; // K
65 xini = [Tsini; CAini; Tini];
66

```

```


67 // TIEMPO
68 tfin = 1500; dt = 1; t = 0:dt:tfin; // s
69
70 // RESOLVER
71 x = ode(xini,0,t,f);
72 disp(size(x))
73 Ts = x(1:N,:); Tsfin = Ts(:, $);
74 CA = x(N+1,:); CAfin = CA($);
75 T = x(N+2,:); Tfin = T($);
76
77 [Tmax,indexTmax] = max(T);
78 tTmax = t(indexTmax);
79
80 // GRÁFICAS
81 scf(1); clf(1);
82 plot(t,CA);
83 xgrid; xlabel('t'); legend('CA',-2,%f);
84
85 scf(2); clf(2);
86 plot(t,T,'r-',t,Ts,'r:',tTmax,Tmax,'ro');
87 xgrid; xlabel('t'); legend('T','Ts',-2,%f);
88
89 scf(3); clf(3);
90 plot(Tsini,'ro-'); // t=0
91 plot(Ts(:, $/2),'ro-'); // t=tfin/2
92 plot(Tsfin,'ro-'); // t=tfin
93 xgrid; xlabel('Sector'); ylabel('Ts');
94
95 disp("El volumen del serpentín es:" + string(Vstot) + "m3")
96 disp("El area del serpentín es:" + string(Astot) + "m2")
97 disp("el volumen del sector es:" + string(Vs) + "m3")
98 disp("el area del sector es:" + string(As) + "m2")
99 disp("la temperatura de salida de los sectores del serpetin son:")
100 for i = 1:N+1
101     if i < N+1 then
102         disp("la temperatura de salida del sector " + string(i) + " es:" + string(Tsfin(i)) + "K")
103     else
104         disp("la temperatura de salida del reactor es:" + string(Tfin) + "K")
105     end
106 end
107 disp("la temperatura maxima en el reactor es:" + string(Tmax) + "K" + " que se da a los " + string(indexTmax) + "s del")
108 disp("el tiempo en el que se alcanza la temperatura maxima es:" + string(tTmax) + "s")
109
110 // Si quiero que el refrigerante sea el necesario para que Tmax = Tfin
111 // ITERACIÓN SOBRE Fs
112 tol = 1e-2; // tolerancia (K)
113 max_iter = 1000; // seguridad
114 iter = 0;
115
116 while iter < max_iter
117
118     // Resolver sistema
119     x = ode(xini,0,t,f);
120
121     Ts = x(1:N,:);
122     CA = x(N+1,:);
123     T = x(N+2,:);
124
125     Tfin = T($);
126     Tmax = max(T);
127
128     // Condición de convergencia
129     if abs(Tmax - Tfin) < tol then

```


```

130     break
131 end
132
133 // Ajuste de Fs
134 if Tmax > Tfin then
135     // sobreenfriando + demasiado refrigerante
136     Fs = Fs - Fs/70;
137 else
138     // falta refrigeración
139     Fs = Fs + Fs/70;
140 end
141
142 iter = iter + 1;
143 end
144
145 disp("Fs óptimo aproximado: " + string(Fs))
146 disp("Iteraciones: " + string(iter))
147 disp("Tmax = " + string(Tmax) + "K" + " Tfin = " + string(Tfin) + "K" + " Diferencia = " + string(abs(Tmax - Tfin)) +

```



(a) Conversión de A vs t



(b) Temperaturas vs t




Figura 8: Temperatura

Resultado:

12.1501.
 El volumen del serpentín es:0.1021018m³
 El area del serpentín es:4.0840704m²
 el volumen del sector es:0.0102102m³
 el area del sector es:0.408407m²
 la temperatura de salida de los sectores del serpetin son:
 la temperatura de salida del sector 1 es:284.72647K
 la temperatura de salida del sector 2 es:286.39235K
 la temperatura de salida del sector 3 es:287.99969K
 la temperatura de salida del sector 4 es:289.5505K
 la temperatura de salida del sector 5 es:291.04671K
 la temperatura de salida del sector 6 es:292.49017K
 la temperatura de salida del sector 7 es:293.88269K
 la temperatura de salida del sector 8 es:295.226K
 la temperatura de salida del sector 9 es:296.52177K
 la temperatura de salida del sector 10 es:297.77162K
 la temperatura de salida del reactor es:328.78124K
 la temperatura maxima en el reactor es:331.28838K que se da a los 1199s del inicio del proceso
 el tiempo en el que se alcanza la temperatura maxima es:1198s
 Fs óptimo aproximado: 5.637D-10
 Iteraciones: 1000
 Tmax = 338.2411K Tfin = 338.22913K Diferencia = 0.0119664K

Pregunta 6: RDMP-4

El equilibrio químico



Factor preexponencial de Arrhenius = 1.75E8 L/(mol · h)

Factor preexponencial de Van't Hoff = 8.25E-22 L/mol

Energía de activación = 62350 J/mol

Entalpía de reacción = -136400 J/mol

se pone en marcha en un reactor discontinuo mezcla perfecta adiabático con 1 mol/L de A y 2 mol/L de B a 300 K. Considerando para la mezcla reaccionante una densidad de 1150 g/L y una capacidad calorífica de 3.8 J/(g · K):

- (A) Obtener la dinámica del proceso
- (B) Determinar cuándo se alcanza el equilibrio, definido por una derivada absoluta de la temperatura inferior a 1E-3 K/h
- (C) Estudiar la influencia de la temperatura inicial en la conversión de equilibrio y en el tiempo necesario para alcanzar el 50 % de conversión

```

1 clear; clc;
2 // RDMP-4.sce
3 // 2 A <=> B
4 // Isotermo
5
6 // SISTEMA DE ECUACIONES DIFERENCIALES
7 function dxdt = f(t,x)
8     // Variables diferenciales
9     CA = x(1)
10    CB = x(2)
11    // Velocidad de reacción
12    // r = rd - ri = kd*CA^2 - ki*CB = kd*CA^2 - kd*CB/Keq

```

```

13     r = kd*(CA^2 - CB/Keq)
14     // Balance de materia para A
15     // d(V*CA)dt = -r*V
16     dCAdt = -r
17     // Balance de materia para B
18     // d(V*CB)dt = r/2*V
19     dCBdt = r/2
20     // Derivadas
21     dxdt(1) = dCAdt
22     dxdt(2) = dCBdt
23 endfunction
24
25 // CONSTANTES
26 T = 420; // K
27 kd0 = 1.4E12; // L/(mol*h)
28 E = 105000; // J/mol
29 Keq0 = 6.9E8; // L/mol
30 H = 63000; // J/mol
31 R = 8.314; // J/(mol*K)
32 kd = kd0*exp(-E/(R*T)) // Ecuación de Arrhenius
33 Keq = Keq0*exp(-H/(R*T)) // Ecuación de Van't Hoff
34
35 // CONDICIONES INICIALES
36 CAini = 5; CBini = 0; // mol/L
37 xini = [CAini; CBini];
38
39 // TIEMPO
40 tfin = 100; dt = 0.1; t = 0:dt:tfin; // h
41
42 // RESOLVER
43 x = ode(xini,0,t,f);
44 CA = x(1,:); CAfin = CA($);
45 CB = x(2,:); CBfin = CB($);
46 disp('Concentracion final de A: '+string(CAfin)+' mol/L');
47 disp('Concentracion final de B: '+string(CBfin)+' mol/L');
48
49 dCAdt = diff(CA)/dt; dCBdt = diff(CB)/dt;
50 dCAkteq = 1E-5; dCBkteq = 1E-5; // mol/(L*h)
51 indexCAeq = find(abs(dCAdt)<dCAkteq,1); indexCBeq = find(abs(dCBdt)<dCBkteq,1);
52 tCAeq = t(indexCAeq); tCBeq = t(indexCBeq);
53 CAeq = CA(indexCAeq); CBeq = CB(indexCBeq);
54 disp('Tiempo de equilibrio para CA: '+string(tCAeq)+' h');
55 disp('Concentracion de equilibrio para CA: '+string(CAeq)+' mol/L');
56 disp('Tiempo de equilibrio para CB: '+string(tCBeq)+' h');
57 disp('Concentracion de equilibrio para CB: '+string(CBeq)+' mol/L');
58 teq = max(tCAeq,tCBeq);
59 disp('Tiempo de equilibrio: '+string(teq)+' h');
60
61 indexCACB = find(CA<CB,1);
62 tCACB = t(indexCACB);
63 CACB = CA(indexCACB);
64 disp('Tiempo en que CA = CB: '+string(tCACB)+' h');
65
66 // GRÁFICAS
67 scf(1); clf(1);
68 plot(t,CA,t,CB);
69 plot(tCAeq,CAeq,'b.',tCBeq,CBeq,'g. ');
70 plot(tCACB,CACB,'bo');
71 xgrid; xlabel('t'); legend('CA','CB',-2,%f);
72
73 scf(2); clf(2);
74 plot(t(1:$-1),abs(dCAdt),t(1:$-1),abs(dCBdt));
75 plot(tCAeq,dCAkteq,'b.',tCBeq,dCBkteq,'g. ');


```




```

76 xgrid; xlabel('t'); legend('|dCAdt|','|dCBdt|',-2,%f);
77 a2 = gca; a2.log_flags = "nl";
78 //! gca da el control de los ejes del grafico
79 //! log_flags = "nl" es para que el eje y sea logaritmico

```



(a) Temperaturas vs t



(b) Temperaturas vs t

Figura 9: Dinámica del proceso

Resultado:

Concentracion final de A: 0.4738786 mol/L
 Concentracion final de B: 2.2630607 mol/L
 Tiempo de equilibrio para CA: 75.6 h
 Concentracion de equilibrio para CA: 0.473956 mol/L
 Tiempo de equilibrio para CB: 69.9 h
 Concentracion de equilibrio para CB: 2.2629811 mol/L
 Tiempo de equilibrio: 75.6 h
 Tiempo en que CA = CB: 3.4 h

Pregunta 7: RDMP-5

El equilibrio químico $2A \rightleftharpoons B$

Factor preexponencial de Arrhenius = $1.4E12$ L/(mol · h)

Factor preexponencial de Van't Hoff = $6.9E8$ L/mol

Energía de activación = 105000 J/mol

Entalpía de reacción = 63000 J/mol

se lleva a cabo a 420 K en un reactor discontinuo mezcla perfecta con 5 mol/L/L iniciales de A.

- (A) Obtener la dinámica del proceso
- (B) Determinar cuándo se alcanza el equilibrio, definido por unas derivadas absolutas de las concentraciones inferiores a $1E-5$ mol/(L · h)
- (C) Localizar el momento en el que las concentraciones de A y B se igualan

```

1 clear; clc;
2 // RDMP-5.sce
3 // A + B <=> C
4 // Adiabático
5

```

```

6  // SISTEMA DE ECUACIONES DIFERENCIALES
7  function dxdt = f(t,x)
8      // Variables diferenciales
9      CA = x(1)
10     CB = x(2)
11     CC = x(3)
12     T = x(4)
13     // Ecuación de Arrhenius
14     kd = kd0*exp(-E/(R*T))
15     // Ecuación de Van't Hoff
16     Keq = Keq0*exp(-H/(R*T))
17     // Velocidad de reacción
18     //  $r = r_d - r_i = k_d \cdot CA \cdot CB - k_i \cdot CC = k_d \cdot CA \cdot CB - k_d \cdot CC / Keq$ 
19     r = kd*(CA*CB - CC/Keq)
20     // Balance de materia para A
21     //  $d(V \cdot CA)dt = -r \cdot V$ 
22     dCAdt = -r
23     // Balance de materia para B
24     //  $d(V \cdot CB)dt = -r \cdot V$ 
25     dCBdt = -r
26     // Balance de materia para C
27     //  $d(V \cdot CC)dt = r \cdot V$ 
28     dCCdt = r
29     // Balance de energía
30     //  $d(V \cdot RHO \cdot CP \cdot T)dt = -H \cdot r \cdot V$ 
31     dTdt = -H*r/(RHO*CP) // Balance de energía
32     // Derivadas
33     dxdt(1) = dCAdt
34     dxdt(2) = dCBdt
35     dxdt(3) = dCCdt
36     dxdt(4) = dTdt
37 endfunction
38
39 // CONSTANTES
40 H = -136400; // J/mol
41 RHO = 1150; // g/L
42 CP = 3.8; // J/(g*K)
43 kd0 = 1.75E8; // L/(mol*h)
44 E = 62350; // J/mol
45 Keq0 = 8.25E-22; // L/mol
46 R = 8.314; // J/(mol*K)
47
48 // CONDICIONES INICIALES
49 CAini = 1; CBini = 2; CCini = 0; // mol/L
50 Tini = 300; // K
51 xini = [CAini; CBini; CCini; Tini];
52
53 // TIEMPO
54 dt = 0.1; tfin = 500; t = 0:dt:tfin; // h
55
56 // RESOLVER
57 x = ode(xini,0,t,f);
58 CA = x(1,:); CAfin = CA($);
59 CB = x(2,:); CBfin = CB($);
60 CC = x(3,:); CCfin = CC($);
61 T = x(4,:); Tfin = T($);
62 XA = 1 - CA/CAini; XAfin = XA($);
63 disp('Concentración final de A: '+string(CAfin)+' mol/L');
64 disp('Concentración final de B: '+string(CBfin)+' mol/L');
65 disp('Concentración final de C: '+string(CCfin)+' mol/L');
66 disp('Conversión final de A: '+string(XAfin)+' mol/L');
67 disp('Temperatura final: '+string(Tfin)+' K');
68

```

```

69
70 dTdt = diff(T)/dt;
71 dTdeq = 1E-3; // K/h
72 indexTeq = find(abs(dTdt)<dTdeq,1);
73 tTeq = t(indexTeq)
74 Teq = T(indexTeq)
75
76 // GRÁFICAS
77 scf(1); clf(1);
78 plot(t,CA,t,CB,t,CC);
79 xgrid; xlabel('t'); legend('CA','CB','CC',-2,%f);
80
81 scf(2); clf(2);
82 plot(t,T,'r-',tTeq,Teq,'r. ');
83 xgrid; xlabel('t'); legend('T',-2,%f);
84
85 XATeq = XA(indexTeq)
86 XAobj = 0.5;
87 indexXAobj = find(XA>XAobj,1);
88 tXAobj = t(indexXAobj)
89
90 scf(3); clf(3);
91 plot(t,XA,'m-',tTeq,XATeq,'m.',tXAobj,XAobj,'mo');
92 xgrid; xlabel('t'); legend('XA',-2,%f);
93
94 scf(4);
95 subplot(211); plot(Tini,XATeq,'mo');
96 xgrid; xlabel('Tini'); ylabel('XATeq');
97 subplot(212); plot(Tini,tXAobj,'go');
98 xgrid; xlabel('Tini'); ylabel('tXAobj');

```

Resultado:

Concentración final de A: 0.1425428 mol/L
 Concentración final de B: 1.1425428 mol/L
 Concentración final de C: 0.8574572 mol/L
 Conversión final de A: 0.8574572 mol/L
 Temperatura final: 326.76365 K






Figura 10: Evolución de la conversión frente al tiempo en el reactor



(a) Temperatura frente tiempo



(b) Conversión frente tiempo

Figura 11: Evolución de la conversión y temperatura en el reactor




Figura 12: Evolución de la conversión y temperatura en el reactor

Pregunta 8: RDMP-6

En un reactor discontinuo mezcla perfecta se realiza el siguiente equilibrio químico durante 5 h a partir de 1 mol/L de A. $A \rightleftharpoons B$

Factor preexponencial de Arrhenius directo = $1.94E15 \text{ h}^{-1}$

Factor preexponencial de Arrhenius inverso = $6.26E19 \text{ h}^{-1}$

Energía de activación directa = 44500 cal/mol

Energía de activación inversa = 59500 cal/mol

Temperatura máxima de operación = 650 K

- (A) Para una conversión dada, obtener la relación entre temperatura y velocidad de reacción
- (B) Determinar la progresión óptima de temperatura
- (C) Determinar la conversión máxima que se puede alcanzar siguiendo una progresión óptima de temperatura

Nota:-

el ultimo es nuevo de este año sustituye los 2 anteriores

```

1  clear; clc;
2  // RDMP-6.sce
3  // A <=> B
4  // Progresión óptima de temperatura
5
6  // (a)
7
8  // CONSTANTES
9  kd0 = 1.94E15; ki0 = 6.26E19; // h-1
10 Ed = 44500; Ei = 59500; // cal/mol
11 R = 1.987; // cal/(mol*K)
12 Tmin = 500.0; dT = 0.1; Tmax = 650.0; T = Tmin:dT:Tmax; // K
13 kd = kd0*exp(-Ed./(R*T)); // Ecuación de Arrhenius directa
14 ki = ki0*exp(-Ei./(R*T)); // Ecuación de Arrhenius inversa
15
16 // CONDICIONES INICIALES
17 CAini = 1; CBini = 0; // mol/L
18
19 XA = 0.0;
20
21 CA = CAini - CAini*XA
22 CB = CBini + CAini*XA
23
24 // Velocidad de reacción
25 // r = rd - ri
26 r = kd*CA - ki*CB;
27 // Velocidad máxima
28 [rmax,indexTopt] = max(r)
29 // Temperatura óptima
30 Topt = T(indexTopt)
31
32 // GRÁFICAS
33 scf(1);
34 plot(T,r,Topt,rmax,'bo');
35 xgrid; xlabel('T'); ylabel('r');
36 a1 = gca;
37 a1.log_flags = "nl" ;
38
39 // (b)
40
41 // SISTEMA DE ECUACIONES DIFERENCIALES
42 function dxdt = f(t,x)
43     // Variables diferenciales
44     CA = x(1)
45     CB = x(2)
46     // Velocidad de reacción
47     // r = rd - ri
48     r = kd*CA - ki*CB
49     // Velocidad máxima
50     [rmax,indexTopt] = max(r)
51     // Temperatura óptima
52     Topt = T(indexTopt)
53     // Balance de materia para A
54     // d(V*CA)dt = -rmax*V
55     dCAdt = -rmax
56     // Balance de materia para B
57     // d(V*CB)dt = rmax*V
58     dCBdt = rmax
59     // Derivadas
60     dxdt(1) = dCAdt

```

```

61     dxdt(2) = dCBdt
62     dxdt(3) = Topt // Almacenar Topt
63 endfunction
64
65 // CONDICIONES INICIALES
66 xini = [CAini; CBini; 0];
67
68 // TIEMPO
69 tfin = 5; dt = 0.01; t = 0:dt:tfin; // h
70
71 // RESOLVER
72 x = ode(xini,0,t,f);
73 CA = x(1,:); CAfin = CA($);
74 CB = x(2,:); CBfin = CB($);
75 Topt = diff(x(3,:))/dt; Toptfin = Topt($);
76 XA = 1 - CA/CAini; XAfin = XA($);
77 disp("La concentracion final de A es: " + string(CAfin) + " mol/L")
78 disp("La concentracion final de B es: " + string(CBfin) + " mol/L")
79 disp("La conversion final de A es: " + string(XAfin*100) + " %")
80 disp("La temperatura optima es: " + string(Toptfin) + " K")
81 // GRÁFICAS
82 scf(2); clf(2);
83 plot(t,XA,'m-');
84 xgrid; xlabel('t'); legend('XA',-2,%f);
85
86 scf(3); clf(3);
87 plot(t(1:$-1),Topt,'r-');
88 xgrid; xlabel('t'); legend('Topt',-2,%f);
89
90 indexTc = find(Topt<Tmax-1E-6,1);
91 tTc = t(indexTc)
92 Tc = Topt(indexTc)
93 disp("La temperatura de control es: " + string(Tc) + " K")
94 disp("El tiempo de control es: " + string(tTc) + " h")
95 plot(tTc,Tc,'ro');
96 ///! probar para diferentes conversiones el grafico 1

```

Nota:-

Revisar la linea de tTC y Tc ademas de revisar el ultimo comentario

Resultado:

La concentracion final de A es: 0.1573513 mol/L
 La concentracion final de B es: 0.8426487 mol/L
 La conversion final de A es: 0.8426487
 La temperatura optima es: 611.29273 K
 La temperatura de control es: 649.85498 K
 El tiempo de control es: 0.97 h






Figura 13: Velocidad de reacción frente a la temperatura







(a) Conversión frente tiempo



(b) Temperatura frente tiempo

Pregunta 9: RDMP-MULT-1

En un reactor discontinuo mezcla perfecta, se llevan a cabo las siguientes reacciones químicas:



- Ⓐ Obtener la dinámica de 500 min si la reacción se pone en marcha con 1 mol/L del compuesto A
- Ⓑ Determinar durante cuánto tiempo:
 - La concentración de A es mayor a 0.5 mol/L
 - La concentración de B está entre 0.15 y 0.20 mol/L
 - La concentración de C crece
 - La concentración de D es predomina

```
1 clear; clc;
2 // RDMP-MULT-1.sce
3 // 1) A => B
```

```

4 // 2) B => C
5 // 3) C => B
6 // 4) B => D
7 // Isotermo
8
9 // SISTEMA DE ECUACIONES DIFERENCIALES
10 function dxdt = f(t,x)
11     // Variables diferenciales
12     CA = x(1)
13     CB = x(2)
14     CC = x(3)
15     CD = x(4)
16     // Velocidades de reacción
17     r1 = k1*CA
18     r2 = k2*CB
19     r3 = k3*CC
20     r4 = k4*CB
21     // Balance de materia para A
22     // d(V*CA)dt = -r1*V
23     dCAdt = -r1
24     // Balance de materia para B
25     // d(V*CB)dt = (r1-r2+r3-r4)*V
26     dCBdt = r1-r2+r3-r4
27     // Balance de materia para C
28     // d(V*CC)dt = (r2-r3)*V
29     dCCdt = r2-r3
30     // Balance de materia para D
31     // d(V*CD)dt = r4*V
32     dCDdt = r4
33     // Derivadas
34     dxdt(1) = dCAdt
35     dxdt(2) = dCBdt
36     dxdt(3) = dCCdt
37     dxdt(4) = dCDdt
38 endfunction
39
40 // CONSTANTES
41 k1 = 0.01; k2 = 0.01; k3 = 0.05; k4 = 0.02; //min-1
42
43 // CONDICIONES INICIALES
44 CAini = 1; CBini = 0; CCini = 0; CDini = 0; // mol/L
45 xini = [CAini;CBini;CCini;CDini];
46
47 // TIEMPO
48 tfin = 500; dt = 0.1; t = 0:dt:tfin; //min
49
50 // RESOLVER
51 x = ode(xini,0,t,f);
52
53 CA = x(1,:); CAfin = CA($)
54 CB = x(2,:); CBfin = CB($)
55 CC = x(3,:); CCfin = CC($)
56 CD = x(4,:); CDfin = CD($)
57
58 // GRÁFICAS
59 scf(1); clf(1);
60 plot(t,CA,t,CB,t,CC,t,CD);
61 xgrid; xlabel('t'); legend('CA','CB','CC','CD',-2,%f);
62
63 CAobj = 0.5;
64 indexA = find(CA > CAobj);
65 tA = dt*length(indexA)
66 disp('tiempo para CA = 0.5 mol/L =' + string(tA) + 'min');

```



```

67 plot(t(indexA),CA(indexA),'b. ');
68
69 CBlo = 0.15; CBup = 0.20;
70 indexB = find(CB > CBlo & CB < CBup);
71 tB = dt*length(indexB)
72 disp('tiempo para 0.15 < CB < 0.20 mol/L =' + string(tB) + 'min');
73 plot(t(indexB),CB(indexB),'g. ');
74
75 dCCdt = diff(CC)/dt;
76 indexC = find(dCCdt > 0);
77 tC = dt*length(indexC);
78 disp('tiempo para dCCdt > 0 =' + string(tC) + 'min');
79 plot(t(indexC),CC(indexC),'r. ');
80
81 indexD = find(CD == max(CA,CB,CC,CD));
82 tD = dt*length(indexD);
83
84 plot(t(indexD),CD(indexD),'c. ');
85 disp('tiempo para CD máximo =' + string(tD) + 'min');
86 indextD = find(CD > CA ,1);
87 disp('tiempo para CD > CA =' + string(t(indextD)) + 'min');

```







Figura 15: Evolución de la concentración frente al tiempo en el reactor

Resultado:

tiempo para $CA = 0.5 \text{ mol/L} = 69.4 \text{ min}$
 tiempo para $0.15 < CB < 0.20 \text{ mol/L} = 65.8 \text{ min}$
 tiempo para $dCCdt > 0 = 98.4 \text{ min}$
 tiempo para $CD \text{ máximo} = 400.2 \text{ min}$
 tiempo para $CD > CA = 99.9 \text{ min}$

Pregunta 10: RDMP-MULT-2

Un reactor discontinuo mezcla perfecta de 1 m³ se pone en marcha a 303 K con 1 mol/L del compuesto A, transcurriendo las siguientes reacciones químicas en disolución acuosa sin generación apreciable de calor durante 2 h:



(A) Considerando que D es el compuesto deseado, determinar cómo debe operar la camisa

```

1  clear; clc;
2  // RDMP-MULT-2.sce
3  // 1) A => B
4  // 2) A => C
5  // 3) B => D*
6  // No adiabático
7
8  // SISTEMA DE ECUACIONES DIFERENCIALES
9  function dxdt = f(t,x)
10     // Variables diferenciales
11     CA = x(1)
12     CB = x(2)
13     CC = x(3)
14     CD = x(4)
15     T = x(5)
16     // Ecuaciones de Arrhenius
17     k1 = exp(-1500/T + 6); //h-1
18     k2 = exp(-4000/T + 12); //h-1
19     k3 = exp(-3000/T + 10); //h-1
20     // Velocidades de reacción
21     r1 = k1*CA
22     r2 = k2*CA
23     r3 = k3*CB
24     // Balance de materia para A
25     // d(V*CA)dt = (-r1-r2)*V
26     dCAdt = -r1 - r2
27     // Balance de materia para B
28     // d(V*CB)dt = (r1-r3)*V
29     dCBdt = r1 - r3
30     // Balance de materia para C
31     // d(V*CC)dt = r2*V
32     dCCdt = r2
33     // Balance de materia para D
34     // d(V*CD)dt = r3*V
35     dCDdt = r3
36     // Temperatura de la camisa
37     if t < tfin/2 then TJ = TJ1;
38         else TJ = TJ2;
39     end
40     // Calor transferido del reactor a la camisa
41     Q = U*A*(TJ-T)
42     // Balance de energía
43     // d(V*RHO*CP*T) = Q
44     dTdt = Q/(V*RHO*CP)
45     // Derivadas
46     dxdt(1) = dCAdt
47     dxdt(2) = dCBdt
48     dxdt(3) = dCCdt
49     dxdt(4) = dCDdt
50     dxdt(5) = dTdt

```

```

51 endfunction
52
53 // CONSTANTES
54 V = 1; // m3
55 A = 5; // m2
56 U = 1000; // kcal/(h*m2*K)
57 RHO = 1000; // kg/m3
58 CP = 1; // kcal/(kg*K)
59 Tcold = 293; Thot = 353; // K
60
61 // Operación de la camisa
62 !!! Esta para una operacion en exclusiva la del caso 1 probar a meter en un bucle for para encontrar la maxima
63 op =1;
64 select op
65     case 1 then TJ1 = Tcold; TJ2 = Tcold; // K
66     case 2 then TJ1 = Thot; TJ2 = Thot; // K
67     case 3 then TJ1 = Tcold; TJ2 = Thot; // K
68     case 4 then TJ1 = Thot; TJ2 = Tcold; // K
69 end
70
71 // CONDICIONES INICIALES
72 CAini = 1; CBini = 0; CCini = 0; CDini = 0; // mol/L
73 Tini = 303; //K
74 xini = [CAini; CBini; CCini; CDini; Tini];
75
76 // TIEMPO
77 tfin = 2; dt = 0.01; t = 0:dt:tfin; //h
78
79 // RESOLVER
80 x = ode(xini,0,t,f);
81 CA = x(1,:); CAfin = CA($);
82 CB = x(2,:); CBfin = CB($);
83 CC = x(3,:); CCfin = CC($);
84 CD = x(4,:); CDfin = CD($);
85 T = x(5,:); Tfin = T($);
86 disp('CAfin ' + string(CAfin) + ' mol/L');
87 disp('CBfin ' + string(CBfin) + ' mol/L');
88 disp('CCfin ' + string(CCfin) + ' mol/L');
89 disp('CDfin ' + string(CDfin) + ' mol/L');
90 disp('Tfin ' + string(Tfin) + ' K');
91
92 // GRÁFICAS
93 scf(1);clf(1);
94 plot(t,CA,t,CB,t,CC,t,CD);
95 xgrid; xlabel('t'); legend('CA','CB','CC','CD',-2,%f);
96
97 scf(2);clf(2);
98 plot(t,T);
99 xgrid; xlabel('t'); legend('T',-2,%f);
100
101 scf(3);
102 bar(op,[CAfin,CBfin,CCfin,CDfin],'stacked');
103 xgrid; xlabel('op'); legend('CA','CB','CC','CD',-2,%f);
104
105
106 /* Probar a hacer graficas para las concentraciones a diferentes T con el bucle for dentro

```






Figura 16: Evolución de la concentración frente al tiempo en el reactor



(a) Evolución de la Temperatura frente al tiempo en el reactor



(b) Concentraciones finales en funcion de la operacion de la camisa

Resultado:

CAfin 0.0049216 mol/L
 CBfin 0.2559964 mol/L
 CCfin 0.0810131 mol/L
 CDfin 0.658069 mol/L
 Tfin 293.00045 K